*-------------------------------------------------------*
* PROGRAMMER:           Liz Uribe			         	*
*                                                     	*
* DATE CREATED:         March 25, 2014			        *
*                                                     	*
* FILE NAME: Print macro example for talk              	*
*														*	
* DESCRIPTION: Uses sashelp.heart to create an example	*
*  demographics table using the prntinit macro			*
*                                                       *
* NOTES: 												*
*														*
* INPUT DATA SETS:                                    	*
*                                                     	*
* OUTPUT DATA SETS:     					          	*
*-------------------------------------------------------*;


%let output_path = H:\work\print macro example for talk; 

data heart;
	set sashelp.heart;
	age = ageatstart;
run;


proc sql noprint;
	select count(status)
	into: total separated by ''
	from heart ;
quit;


proc means data = heart mean std min max nmiss;
var  age height weight cholesterol diastolic systolic ;
output out = summary 	  mean = mean_age mean_ht mean_wt mean_chol mean_dia mean_sys  
						  std = std_age std_ht std_wt std_chol std_dia std_sys 
						  min = min_age min_ht min_wt min_chol min_dia min_sys 
					 	  max = max_age max_ht max_wt max_chol max_dia max_sys 
						  nmiss = nmiss_age nmiss_ht nmiss_wt nmiss_chol nmiss_dia nmiss_sys ;
run;


data summary2 (drop=mean: std: min_: max: nmis: _TYPE_);
	set summary;
	length avestd_age avestd_ht avestd_wt avestd_chol avestd_dia avestd_sys $12.;

	array mean {*} mean_age mean_ht mean_wt mean_chol mean_dia mean_sys  ;
	array std {*} std_age std_ht std_wt std_chol std_dia std_sys  ;
	array avestd {*} $ avestd_age avestd_ht avestd_wt avestd_chol avestd_dia avestd_sys  ;

	array min {*} min_age min_ht min_wt min_chol min_dia min_sys  ;
	array max {*} max_age max_ht max_wt max_chol max_dia max_sys ;
	array minmax {*} $10. minmax_age minmax_ht minmax_wt  minmax_chol minmax_dia minmax_sys;

	array nmis {*} nmiss_age nmiss_ht nmiss_wt nmiss_chol nmiss_dia nmiss_sys ;
	array miss {*} $ miss_age miss_ht miss_wt miss_chol miss_dia miss_sys ;

	do i=1 to dim(mean);
		avestd[i] = strip(put(mean[i], 5.2))  || " (" || strip(put(std[i], 5.1)) || ")"; 
		minmax[i] = "(" || strip(put(min[i],5.)) || ", " || strip(put(max[i],5.)) || ")";
		miss[i] = strip(put(nmis[i], 3.)); 
	end;
run;


proc transpose data=summary2 out=summary3; 
var avestd_age minmax_age miss_age 
    avestd_ht minmax_ht miss_ht
	avestd_wt minmax_wt miss_wt 
	avestd_chol minmax_chol miss_chol
	avestd_dia minmax_dia miss_dia
	avestd_sys minmax_sys miss_sys;
run;

data summary4;
	set summary3;
	if _N_ < 13 then i = _N_+3;
	else i = _N_ + 12;
run;


*categorical stats;
%macro transpose(var);

	proc freq data=heart;
		tables &var / missing out = freq_&var;
	run; 

	data freq2_&var (keep= &var COL1);
		set freq_&var;
		COL1 = strip(put(COUNT,5.)) || " (" || strip(put(PERCENT,5.1)) || "%)";
		if missing(&var) then &var = "Missing";
	run;
%mend;

%transpose(status); %transpose(sex); %transpose(smoking_status);

data status;
	set freq2_status;
	i = _N_;
run;

data gender;
	set freq2_sex;
	i = _N_ + 15;
run;

data smoke;	
	set freq2_smoking_status;
	if Smoking_status = "Non-smoker" then i = 19;
	else if Smoking_status = "Light (1-5)" then i = 20;
	else if Smoking_status = "Moderate (6-15)" then i = 21;
	else if Smoking_status = "Heavy (16-25)" then i = 22;
	else if Smoking_status = "Very Heavy (> 25)" then i = 23;
	else if Smoking_status ="Missing" then i = 24; 
run;
proc sort data = smoke; by i; run;


data all;
	set summary4 status gender smoke;
	by i;
	newtable = 1;
run;


*Cause of death for those who died;
proc sql noprint;
	select count(status)
	into: dead separated by ''
	from heart 
	where status = "Dead";
quit;

proc freq data = heart(where = (status = "Dead"));
	tables deathcause / out = freq_deathcause outpct sparse;
run;

data freq2_deathcause (keep = deathcause n_per newtable i);
	set freq_deathcause;
	n_per = strip(put(COUNT,5.)) || " (" || strip(put(PERCENT,5.1)) || "%)";
	i = _N_;
	newtable = 2;
run;


data all2(drop=_NAME_ sex smoking_status status deathcause);
	set all freq2_deathcause;
	by newtable i;
run;

*combine everything;
data labels;
input newtable i shortname $42.;
cards;
1 1 Alive
1 2 Dead
1 3 Missing
1 4 Mean (Std)
1 5 (Min, Max)
1 6 Missing
1 7 Mean (Std)
1 8 (Min, Max)
1 9 Missing
1 10 Mean (Std)
1 11 (Min, Max)
1 12 Missing
1 13 Mean (Std)
1 14 (Min, Max)
1 15 Missing
1 16 Female
1 17 Male
1 18 Missing
1 19 Non-Smoker
1 20 Light (1-5)
1 21 Moderate (6-15)
1 22 Heavy (16-25)
1 23 Very Heavy (> 25)
1 24 Missing
1 25 Mean
1 26 (Min, Max)
1 27 Missing
1 28 Mean
1 29 (Min, Max)
1 30 Missing

2 1 Cancer
2 2 Cerebral Vascular Disease
2 3 Coronary Heart Disease
2 4 Other
2 5 Unknown 
;
run;

data headers;
input newtable i name $42.;
cards;
1 0.5 Status
1 3.5 Age 
1 6.5 Height
1 9.5 Weight 
1 12.5 Cholesterol
1 15.5 Gender
1 18.5 Smoking Status  
1 24.5 Diatolic Blood Pressure
1 27.5 Systolic Blood Pressure 
;
run;


data combine ;
	merge labels headers all2;
	by newtable i;

	if missing(name) then name = '   ' || strip(shortname);

	if not missing(shortname) and newtable = 1 then do;
		if missing(col1) then col1 = "0 (0%)";
	end;
run;





options nocenter nodate;
%INCLUDE 'K:\DMC\SAS LIBRARY\PRINTDRV\PRNTINIT.SAS';

data _null_;
set combine end = eof;
by newtable i;

%PRNTINIT(DRIVER=RTF,PRINT="&output_path\Example demographics table.rtf",
			FONT=ARIAL,FONTSIZE=8, ORIENT=portrait, LS=340,PS=85,TJUST=LEFT,MARGINS=LEFT:1);

if newtable = 1 then do;
	
	if first.newtable then do;
		%print("Example Demographics Table", just = left, style = bold);
		%print(2, proc = newline);
		%print(proc = table, column = 1 60 90, lines = all:single);
		%print(proc = newrow);

		%print("Variable",	column = 1, style = bold, just = center, lines = bottom:none);
		%print("Subjects",	column = 2, style = bold, just = center, lines = bottom:none);
		%print(proc = newrow);
		
		%print( 				column = 1);
		%print("(N = &total)", 	column = 2, just = center);
		%print(proc = newrow);
	end;

	if not last.newtable then do;
		if missing(col1) then do;
			%print(name, 		column = 1, style = bold, lines = top:single);
			%print(				column = 2, just = center, lines = top:single);
			%print(proc = newrow);
		end;
		else if not missing(col1) then do;
			%print(name, 	column = 1, lines = bottom:none);
			%print(col1, 		column = 2, just = center, lines = bottom:none);
			%print(proc = newrow);
		end;
	end;

	if last.newtable then do;
		%print(name, 	column = 1);
		%print(col1, 		column = 2, just = center);
		%print(proc = endtable);

		%print(proc = newline);
		%print("Comments can be added");
		%print(3, proc = newline);
	end;
end;

if newtable = 2 then do;
	if first.newtable then do;
		%print("For Those who Died: Causes of Death", just = left, style = bold);
		%print(2, proc = newline);
		%print(proc = table, column = 1 60 90, lines = all:single);
		%print(proc = newrow);

		%print("Cause of Death",	column = 1, style = bold, just = center, lines = bottom:none);
		%print("Count",	column = 2, style = bold, just = center, lines = bottom:none);
		%print(proc = newrow);
		
		%print( 				column = 1);
		%print("(N = &dead)", 	column = 2, just = center);
		%print(proc = newrow);
	end;

	if not eof then do;
		%print(name, 		column = 1, lines = bottom:none);
		%print(n_per, 		column = 2, just = center, lines = bottom:none);
		%print(proc = newrow);
	end;

	if eof then do;
		%print(name, 	column = 1);
		%print(n_per, 	column = 2, just = center);
		%print(proc = endtable);

		%print(proc = newline);
		%print("Comments can be added");
		%prntstop;	
	end;		
end;

run;

